Gramática

Representa la *definición formal* de la **sintaxis** de un lenguaje.

Posee un conjunto de reglas que especifican las reglas de escritura para formar estructuras válidas en un lenguaje.

Metalenguajes

Gramática formal destinada a la descripción de un lenguaje.

Existen dos metalenguajes comúnmente utilizados :

- **BNF (Backus-Naur-Form)**
- **Diagramas sintácticos**

BNF (Backus - Naur Form)

Notación desarrollada por los especialistas **Backus** y **Naur** para definir lenguaje Algol 60

Metasímbolos:

- **< >:** indica símbolo NO-TERMINAL o meta variable
- **::=** : "Se define como"
- **|**: " o"
- **{ }$_n$**: Repetición. Mínimo **n** veces.
- **identificador**: Palabra reservada, constante o carácter TERMINAL.
- **[ ]** : indica opcionalidad, puede o no ser incluído en el lenguaje.

EJEMPLO: BNF NÚMERO REAL, IDENTIFICADOR

<real> ::= <secuencia> .<secuencia>

<secuencia> ::= <dígito> {<dígito>}$_0$

<dígito> ::= 0 | 1 | 2 | 3 | ...8 | 9

<identificador> ::= **<letra>** **{<letra> | <dígito>}$_0$**

<letra> ::= A | B | C | ...Y | Z | Aa | b | c...y | z

Ejemplo BNF: Expresión aritmética

Expresión:



<exp>::= <término> | <término>+<término> | <término>-<término>

<término>::= <factor> | <factor> * <factor> | <factor> / <factor>

<factor>::= <identificador> | <constante> | (<exp>)

BNF Sentencia For de Pascal

<s-for>::= For <identificador> := <intervalo> do <sentencia>

<intervalo>::= <inicial> to <final> | <inicial>downto < final>

<inicial> ::= <exp>

<final> ::= <exp>

BNF Recursivas

<entero>::= <dígito> | <dígito><entero>

<dígito>::= 0|1|2|..|9

.....

<real> ::= <secuencia> • <secuencia>

<secuencia>::= <dígito>| <dígito> <secuencia>

BNF Recursivas

<identificador> ::= <letra> | <letra> <secuencia>

<secuencia> ::= <carácter> | <carácter><secuencia>

<carácter> ::= <letra> | <dígito>

BNF Sentencia Pascal

<sentencia> ::= <simple> | <compuesta>

<simple>::= <s-asig> | <s-inv> | <s-dec> | <s-iter>

<compuesta>::= **Begin**<grupo-sentencia> **End**

<grupo-sentencia>::= <simple>

| <simple> ;<grupo-sentencia>

BNF EXPRESIÓN ARITMÉTICA

<exp>::= <término> | <término> **+<exp>**

| <término> **- <exp>**

<término>::= <factor> | <factor> *** <término>**

| <factor> **/ <término>**

<factor>::= <identificador> | <constante>

| **(**<exp>**)**

BNF Expresión Recursiva

a + b*c



BNF Expresión Recursiva

a - b - c



BNF Expresión Recursiva

(a +b) *c



BNF Sentencias Pascal

<s-While>::= **While** <exp B> **do** <sentencia>

<s-If>::= **If** <exp B> **then** <sentencia>

| **If** <exp B> **then** <sentencia> **else** <sentencia>

BNF Sentencias C

```
do
{printf("Número ");
  scanf("%d",&n);
}while (n<=0);
```

<do-while>::= **do** <sentencia> **while** <exp B>

<sentencia>::= <sentencia simple> | <sentencia compuesta>

<sentencia compuesta>::="{"{<sentencia simple>}$_1$" }"

BNF SENTENCIAS C

```
switch (x)
{case 1: cout<<"es UNO";break;
  case 2:
  case 3: cout <<"es dos o tres";break;
  default : cout <<"es distinto de 1,2 ó 3";
}
```

<switch>::= switch (<exp>) {<secuencia>}

<secuencia> ::= <caso> | <caso> <secuencia> | <caso><defecto>

<caso> ::= case <constante>:<sentencias>;

| case <constante> : <sentencias>; break;

<defecto> ::= default :<sentencias>;

<constante>::= enteros | caracteres

Diagramas Sintácticos

Constituyen un ***método de descripción*** de lenguajes, equivalente a la BNF, originalmente propuesto por N. Wirth. para definir sintáxis de Pascal.

Diagramas Sintácticos

<S>::=<v_1> | <v_2> ··· | <v_n>



Cada ocurrencia de un símbolo terminal corresponde al diagrama:



Cada ocurrencia de un símbolo no terminal corresponde al diagrama



Diagramas Sintácticos

Una producción de la forma:

$$\langle S \rangle ::= \{\langle A \rangle\}_0$$

corresponde al diagrama:

(mientras)



Una producción de la forma:

$$\langle S \rangle ::= \langle x \rangle\{\langle x \rangle\}_0$$

corresponde al diagrama:

(repetir)



Diagramas Sintácticos
Pascal

Identificador:



Letra:



Dígito:



Diagramas Sintácticos
Pascal

Número Entero:



Número Real:



Diagramas Sintácticos
Pascal

Sentencia:



Sentencia Simple:



Diagramas Sintácticos
Pascal

Sentencia Compuesta:



Diagramas Sintácticos Pascal

Sentencia Asignación:



Sentencia if:



Sentencia while:



Ejercicio: BNF sentencia Select SQL

<sentencia select>::= **SELECT** [**ALL** | **DISTINCT**] <lista de selección>

FROM { {<nombre de tabla> | <nombre de vista> } [<nombre de sinónimo>]},…

[**WHERE** <condición de búsqueda>]

[**GROUP BY** {<nombre de columna> | <número de columna>},…]

[**HAVING** <condición de búsqueda>]

[{**INTERSECT** | **MINUS** | **UNION** [**ALL**] } <sentencia select>]

[[**ORDER BY** {{<nombre de columna | <número de columna>} [**ASC** | **DESC**] },…] |

[**FOR UPDATE OF** <nombre de columna>,…]]

<lista de selección>::= {***** | {<expresión> | <nombre de tabla>**.**<nombre de columna> | <nombre de columna> |

<nombre de tabla>**.*** | <nombre de vista>**.**<nombre de columna> |

<nombre de vista>**.*** | <nombre de sinonimo>**.**<nombre de columna> | <nombre de sinonimo>**.***},…}

// **DISTINCT** indica que si existen filas idénticas, sólo se mostrará una de ellas.

¿Son válidas las instrucciones:

- **SELECT ALL col1,col2,col3 FROM mitabla**
- **SELECT col1,col2,col3 FROM mitabla**

Ejercicio: Diagramas Sintácticos, select sql



Son válidas las instrucciones:

- **SELECT DISTINCT col1 FROM mitabla**
- **SELECT col1,col2 FROM mitabla WHERE col2 = 0**

EJERCICIO:

Verifique la validez de las siguientes instrucciones SQL:

1. SELECT S# FROM S WHERE Ciudad=`PARIS' AND Situación > 20;

2. SELECT S.*,P.* FROM PROVEEDORES, PIEZAS WHERE PROVEEDORES.CIUDAD= PIEZAS.CIUDAD;

3. SELECT P_CIUDAD, COLOR (DISTINT) FROM PIEZAS ;

4. SELECT COLOR,P_CIUDAD FROM PIEZAS WHERE COLOR='VERDE' OR COLOR='ROJO';

PROBLEMA:

Una expresión sintáctica, mediante BNF, como

<fecha> ::= <d><d>/<d><d>/<d><d><d><d>

puede tener dos *interpretaciones semánticas*; por ejemplo:

09/04/2012 se entiende como

- 9 de Abril de 2012 en Chile
- 4 de Septiembre de 2012 en EEUU.